

## MAX3266C/D I/O Model

SPICE I/O Macromodels aid in understanding signal integrity issues in electronic systems. Most of Maxim's High Frequency/Fiber Communication ICs utilize input and output (I/O) circuits with Current Mode Logic (CML), Positive Emitter Coupled Logic (PECL), and Low Voltage Differential Signal (LVDS) formats to transfer data into and out of an IC. These models are based on simplified circuit expressions that may include replacement of active circuit elements with ideal controlled voltage and current sources. As such, simulation with macromodels should be treated as 'typical' performance and not relied upon as final proof-of-design. Use of macromodel descriptions is not a substitute for worst-case design analysis, nor for testing real circuits over temperature, supply, and other operating limits.

The output format is provided as ASCII text netlists suitable for generic SPICE. This format is compatible with most versions of SPICE such as PSPICE and HSPICE. Additional information is found in HFAN 6.1 *Input/Output Models for Maxim Fiber Components*.

To extract the circuit netlists using the Adobe Acrobat Reader follow these instructions. Select the "Text Select Tool" by clicking the left mouse button on this icon of the menu bar (a capital T with a small dashed box to the lower right). Highlight the desired netlist text with the cursor. Use the copy command from the edit menu to capture the selected lines. Then paste the selected lines into a text file editor and save the file with an extension compatible with the simulator.

Version A, August 11, 2003

## MAX3266 Transimpedance Amplifier




Figure 1. Output signal buffer for the MAX3266 used in dice form. For bondwires estimate 0.9nH/mm and 0.04Ohms/mm.

## Notes:

The schematic on the previous page represents the input stage of the Maxim MAX3266C/A Transimpedance amplifiers. The input stage is not modeled since the electrical model is not considered an adequate representation for the analysis of the input optical signal.

The netlist is in SPICE 2g6 format. Since nodes in SPICE 2g6 can only be numbers, the output signals are 2101 and 2102 and the input signals to the die are 1001 and 1002. Comments in the netlist identify the correspondence between the signal names and the node numbers. The netlists are in SPICE 2g6 format and are compatible with PSPICE and HSPICE. It has been simulated on a generic SPICE simulator and PSPICE.

**The Output Stage:** The input stage is comprised of two subcircuits OUTPKG and DRV\_OUT. RB1, RB2, CB1 and CB2 can be adjusted to slow or speed up the output signals

**Text File Format:** This model is shipped in “pdf” format. Models and netlists can be copied to text format in the Acrobat Reader by holding the left mouse button on the “Text Select Tool.” Then the user can “select” what netlist and/or subcircuit with the mouse. Then use the copy command from the “edit” menu to capture the selected lines. These lines can then be “pasted” into the user’s text file.

## Circuit Netlist – Output Circuit

```
INPUT - MAX3266C/D OUTPUT CIRCUIT
*
* THIS IS THE TYPICAL CML OUTPUT OF THE MAX3266 without package.
*
.OPTIONS ACCT NOMOD NOPAGE LIMPTS=10000 RELTOL=.001
.WIDTH OUT=80
.TEMP 26
* TYPICAL DIE TEMP = 25C + 0.143W*(6.7C/W) = 26C
.OP
.TRAN 5PS 3000PS
*
* CONVENTIONS VCC = 101, VEE = 102, + OUT = 2000, - OUT = 2002
*
VCC 101 0 DC 3.3
RTERM1 2001 101 100
RTERM2 2002 101 100
RLOAD1 2001 2002 100
CLOAD1 2001 101 0.20P
CLOAD2 2002 101 0.20P
CLOAD3 2001 2002 0.05P

*XPK1 2001 2002 2010 2011 0 0 0 OUTPKG
XCIROUT 2001 2002 101 OUTDRV
*
.SUBCKT OUTDRV 71 72 101
VINP 2 0 PULSE (1.25 0.95 0.2N 0.300N 0.300N 0.5000N 1.600N)
VINN 3 0 PULSE (0.95 1.25 0.2N 0.300N 0.300N 0.500N 1.600N)
*
RB1 2 22 200
CB1 22 0 .3p
* Adjusted to match waveform of data sheet
RB2 3 32 200
CB2 32 0 .3p
* Adjusted to match waveform of data sheet
*
XQ1 101 22 64 0 N11A03_4
XQ2 71 64 66 0 N11A04_8

XQ3 101 32 65 0 N11A03_4
XQ4 72 65 67 0 N11A04_8

RE1 66 63 26
RE2 67 63 26
*

RC1 71 101 50
RC2 72 101 50

*
IB1 63 0 16M
IB2 64 0 2.5M
IB3 65 0 2.5M

XPAD1 71 101 0 PAD3
XPAD2 72 101 0 PAD3

.ENDS OUTDRV
*
```

\*

\*

\*

\*\* BEGINNING OF PROCESS LIB

\*

```
.SUBCKT N11A03_4 1 2 3 21
CP1EPI 10 12 14.156F
CP1P2 12 3 11.941F
CTRENCH 1 20 25.820F
RBX 2 12 29.695 TC=2.556M
RCX 1 10 14.554 TC=2.576M,499.699N
RCI 10 11 3.639 TC=2.576M,499.699N
REX 13 3 4.519 TC=192.895U
RSUB 20 0 4.155K
QP 20 10 12 20 TXP 4 OFF
QN 11 12 13 11 TX 4
*XREPORT1 0 REPORTERL1N3
*XREPORT2 0 REPORTERL1N9
.MODEL TX NPN( IS=6.897E-018 XTI=3 EG=1.155 BF=331.407 BR=36 XTB=0
+ VAF=100 VAR=2 NF=1.018 NR=1.018 NE=2 NC=1.560 IKF=22.765M
+ IKR=556.875U ISE=4.455E-016 ISC=2.005E-030 RB=29.695 RBM=22.271
+ IRB=2.005M CJE=10.919F MJE=463M VJE=1.040 FC=990M CJC=7.144F
+ MJC=400M VJC=890M TF=2.631P TR=19N XTF=100 VTF=5 ITF=65.510M PTF=5
+ KF=9.000E-016 AF=1.500 )
.MODEL TXP PNP( IS=5.724E-020 CJE=7.216E-017 MJE=400M VJE=890M
+ CJC=5.346F MJC=460M VJC=790M BF=10K BR=55.381M TF=1N FC=900M )
.ENDS N11A03_4
```

```
.SUBCKT N11A04_8 1 2 3 21
CP1EPI 10 12 37.431F
CP1P2 12 3 29.064F
CTRENCH 1 20 59.018F
RBX 2 12 9.232 TC=2.542M
RCX 1 10 4.789 TC=2.490M,455.574N
RCI 10 11 1.197 TC=2.490M,455.574N
REX 13 3 1.523 TC=303.554U
RSUB 20 0 1.722K
QP 20 10 12 20 TXP 8 OFF
QN 11 12 13 11 TX 8
*XREPORT1 0 REPORTERL1N5
*XREPORT2 0 REPORTERL1N10
.MODEL TX NPN( IS=1.150E-017 XTI=3 EG=1.155 BF=315.986 BR=36 XTB=0
+ VAF=100 VAR=2 NF=1.018 NR=1.018 NE=2 NC=1.560 IKF=37.942M
+ IKR=928.125U ISE=7.425E-016 ISC=3.341E-030 RB=18.464 RBM=13.848
+ IRB=3.341M CJE=18.181F MJE=463M VJE=1.040 FC=990M CJC=10.660F
+ MJC=400M VJC=890M TF=2.668P TR=19N XTF=100 VTF=5 ITF=107.589M PTF=5
+ KF=9.000E-016 AF=1.500 )
.MODEL TXP PNP( IS=8.640E-020 CJE=1.077E-016 MJE=400M VJE=890M
+ CJC=7.351F MJC=460M VJC=790M BF=10K BR=60.796M TF=1N FC=900M )
.ENDS N11A04_8
```

```
.SUBCKT PAD3 1 3 21
CPAD 1 10 87.228F
REPI 10 20 250.111M TC=4.800M,5U
CTRENCH 21 20 22.394F
DS 21 20 DSUB
RS 3 21 369.115
*XREPORT1 0 REPORTERL1N53
.MODEL DSUB D( IS=12.656F CJO=885.938F M=500M VJ=450M )
.ENDS PAD3
```

```
.SUBCKT DE381011 1 2 21
```

```
CP1EPI 1 4 132.715F
QD 5 4 1 5 QESD
RS 4 2 2.024 TC=2.615M,1.746U
RSUB 5 21 2.318K
CTRENCH 2 5 38.731F
*XREPORT1 0 REPORTERL1N49
*XREPORT2 0 REPORTERL1N50
.MODEL QESD PNP( IS=1.080E-017 NF=1.050 BF=800M BR=600U CJE=220.280F
+ VJE=640M MJE=330M CJC=75.512F VJC=790M MJC=460M )
.ENDS DE381011
```

```
.SUBCKT OUTPKG 101 102 201 202 401 402 403
```

```
*
* resistors
*
```

```
RB01 201 301 0.25
RB02 202 302 0.25
```

```
*
* inductors
*
```

```
LLAP_1_3 101 301 1.2N
LLAP_2_4 102 302 1.3N
K02_03 LLAP_1_3 LLAP_2_4 0.294
```

```
*LB03 PADT PADBOT 23P
```

```
*
* capacitors
*
```

```
C01 101 403 143F
C02 102 403 143F
```

```
*
* mutual capacitors
*
```

```
C01_02 101 102 43.800F
.ENDS OUTPKG
```

```
*
.PRINT TRAN V(2001) V(2002)
*.PROBE
*
.END
```